Sample Tank Problem

Setup

Consider the following three tank setup, where the volume of water in each tank is 1 kL (kilolitre). Fresh water flows into Tank 1 at a rate of k1 kL/min, and water is dumped out of Tank 2 at a rate of k1 kL/min. Water circulates through the system with the rates as indicated (these are designed so the volume of water in each tank remains constant). (This drawing was made by selecting “Graphics” in the menu, then choosing “New Graphic,” and then also selecting “Drawing Tools” from under the “Graphics” menu.

In[1]:=



We may model this situation as a system of linear differential equations.
Let x1, x2, x3 be the masses of a solute (e.g. salt or chlorine) present in Tanks 1,2,3 (respectively) after t seconds.
Assume the initial solute masses m1, m2, m3 in each tank.

Our system of linear differential equations may be written as

$$\begin{cases} x1' = & -k2\,x1 \quad + (k2-k1)\,x3 \\ x2' = & k2\,x1 - k1\,x2 - (k2-k1)\,x3 \\ x3' = & (k2-k1)\,x2 - (k2-k1)\,x3, \end{cases}$$

and this can be rewritten as the matrix differential equation X’ = AX, X(0) = X0, where A is defined below. The solution is given by X = e^(At) X0.

Minimal analysis. (A good start)

Here are solutions for a few specific cases.

In[2]:=
```
{k1, k2} = {.01, .02};   (*Flow rates*)
{m1, m2, m3} = {50, 20, 50};   (*Initial solutes*)

a = {{-k2, 0, k2 - k1}, {k2, -k1, -(k2 - k1)}, {0, k2 - k1, -(k2 - k1)}};
X0 = {{m1}, {m2}, {m3}};
MatrixExp[a t].X0 // MatrixForm
```

Out[6]//MatrixForm=
```
( 50 × ((0.494627 + 0. i) × ((0. + 0. i) + (0.369579 + 9.54559 × 10^-18 i) × ((0. + 0. i) + 1. e^-0.
  50 × ((0.415134 + 0. i) × ((0. + 0. i) + (0.369579 + 9.54559 × 10^-18 i) × ((0. + 0. i) + 1. e^-0.0045
  50 × ((0.76355 + 0. i) × ((0. + 0. i) + (0.369579 + 9.54559 × 10^-18 i) × ((0. + 0. i) + 1.
```

In[7]:=
```
{k1, k2} = {.03, .02};   (*Flow rates*)
{m1, m2, m3} = {40, 30, 50};   (*Initial solutes*)

a = {{-k2, 0, k2 - k1}, {k2, -k1, -(k2 - k1)}, {0, k2 - k1, -(k2 - k1)}};
X0 = {{m1}, {m2}, {m3}};
Chop[FullSimplify[MatrixExp[a t].X0]] // MatrixForm  (*The solution,
 as a function of t.  Chop replaces numbers that are close to 0 by 0. *)
```

Out[11]//MatrixForm=

$$\begin{pmatrix} e^{-0.0245982\,t}\left(-11.651\,e^{0.0337946\,t} + 51.651\,\mathrm{Cos}[0.00688173\,t] + 11.2864\,\mathrm{Sin}[0.00688173\,t]\right) \\ e^{-0.0245982\,t}\left(2.73361\,e^{0.0337946\,t} + 27.2664\,\mathrm{Cos}[0.00688173\,t] + 151.933\,\mathrm{Sin}[0.00688173\,t]\right) \\ e^{-0.0245982\,t}\left(34.0169\,e^{0.0337946\,t} + 15.9831\,\mathrm{Cos}[0.00688173\,t] + 40.7346\,\mathrm{Sin}[0.00688173\,t]\right) \end{pmatrix}$$

While this sort of analysis give correct numbers, it is difficult to get an understanding of the situation (for several reasons). However, getting these specific solutions are often an important first step in understanding a more general problem.

A much better analysis.

In[12]:=
```
Clear[a, k1, k2, m1, m2, m3]; (*Now, k1 is just k1, as opposed to .03.*)
a[k1_, k2_] := {{-k2, 0, k2 - k1}, {k2, -k1, -(k2 - k1)}, {0, k2 - k1, -(k2 - k1)}};
X0 = {{m1}, {m2}, {m3}};
MatrixExp[a[k1, k2] t].X0 // MatrixForm
```

Out[15]//MatrixForm=

$$\left(\begin{array}{l} (-k1+k2)^2\, m2\, \text{RootSum}\left[k1^2\, k2 - k1\, k2^2 + k1\, k2\, \#1 - 2\, k2^2\, \#1 - 2\, k2\, \#1^2 - \#1^3\, \&,\ \frac{e^{t\,\#1}}{-k1\, k2 + 2\, k2^2 + 4\, k2\, \#1 + 3\, \#} \right. \\ (k1-k2)\, m3\, \text{RootSum}\left[k1^2\, k2 - k1\, k2^2 + k1\, k2\, \#1 - 2\, k2^2\, \#1 - 2\, k2\, \#1^2 - \#1^3\, \&,\ \frac{e^{t\,\#1}\, \#1}{-k1\, k2 + 2\, k2^2 + 4\, k2\, \#1 + 3\, \#1^2} \right. \\ k2\, (-k1+k2)\, m1\, \text{RootSum}\left[k1^2\, k2 - k1\, k2^2 + k1\, k2\, \#1 - 2\, k2^2\, \#1 - 2\, k2\, \#1^2 - \#1^3\, \&,\ \frac{e^{t\,\#1}}{-k1\, k2 + 2\, k2^2 + 4\, k2\, \#1 + 3} \right. \end{array}\right.$$

Our first attempt at a general solution did not yield anything very helpful. Let's investigate the characteristic polynomial to see what our solutions will look like.

In[16]:=
```
CharacteristicPolynomial[a[k1, k2], z]
```

Out[16]= $k1^2\, k2 - k1\, k2^2 + k1\, k2\, z - 2\, k2^2\, z - 2\, k2\, z^2 - z^3$

Hmm... this is tough too. What about looking at the roots for specific values within the range that would probably be relevant to the problem?

In[17]:=
```
Manipulate[
 Roots[CharacteristicPolynomial[a[k1, k2], z] == 0, z],
 {k1, .001, .5}, {k2, .001, .5}]
```

Out[17]=

k1

k2

z == -0.0969023 - 0.0675305 i || z == -0.0969023 + 0.0675305 i || z == -0.0221955

Upon inspection, it appears we will have two complex eigenvalues with negative real part. This means we will have solutions of the form e^(-** t) (c1 cos(** t) + c2 sin(** t)). The third eigenvalue appears to be real; it is positive if k1 > k2, and it is negative if k2 > k1. But, we implicitly assumed that k2 > k1, since k2-k1 is the flow rate from Tank 2 to 3, and from Tank 3 to 1. Therefore, the third eigenvalue will be negative, giving us a solution of the form e^(-** t).

Hence, we expect for our final solution to exhibit some sort of oscillation while converging towards the long-term solution of 0 solvent in each tank.

Here are graphs of the solution curves for various parameters with the specific m1,m2,m3=(50,20,30). Notice that we should disregard the cases where k1>k2.

In[18]:=
```
soln[k1_, k2_, t_] := MatrixExp[ a[k1, k2] t].{{50}, {20}, {30}};

Manipulate[
 Plot[ {soln[k1, k2, t]⟦1, 1⟧, soln[k1, k2, t]⟦2, 1⟧ , soln[k1, k2, t]⟦3, 1⟧} ,
  {t, 0, 500}, PlotRange → {0, 50}, PlotLegends → {"x1", "x2", "x3"}],
 {k1, .001, .05}, {k2, .001, .05}]
(* [[i,j]] is being used to obtain the (i,j) entry of the matrix.*)
```

Out[19]=



Here’s a slightly nicer implementation, and we include the ability to modify initial quantities.

In[20]:=
```
table[k1_, k2_, t_, m1_, m2_, m3_] :=
  Table[(MatrixExp[ a[k1, k2] t].Transpose[{{m1, m2, m3}}])[[i, 1]], {i, 1, 3}];

Manipulate[
 Plot[ Evaluate[table[k1, k2, t, m1, m2, m3] ], {t, 0, 500},
  PlotRange → {0, 50}, PlotLegends → {"x1", "x2", "x3"}],
 {k1, .001, .05}, {k2, .001, .05}, {m1, 0, 50}, {m2, 0, 50}, {m3, 0 , 50}]
```

Out[21]=