The following is a function SymbolicRR that makes row reduction more compatible with symbolic algebra.
Author: Corbett Redden (corbett.redden@liu.edu)

```
SymbolicRR[matrix_, symbols_:{}] :=
 FixedPoint[Function[mtrx, Simplify[RowReduce[mtrx,
     ZeroTest → (Apply[Or, Table[! FreeQ[#, symb], {symb, symbols}]] ||
         PossibleZeroQ[#] &)]]], matrix, 10]
```

```
SymbolicRR::usage =
  "SymbolicRR[matrix, {var1, var2, ...}] row reduces a matrix without
   assuming expressions involving variables are non-zero.";
```

In ordinary RowReduce, *Mathematica* will implicitly assume all expressions involving variables are invertible. While we get the "desired answer" in this first case, ...

$$\text{RowReduce}\left[\begin{pmatrix} 1 & 1 & 3 & x \\ 2 & 0 & 2 & y \\ 1 & 1 & 5 & z \end{pmatrix}\right] \text{ // MatrixForm}$$

$$\begin{pmatrix} 1 & 0 & 0 & \frac{1}{2}(x+y-z) \\ 0 & 1 & 0 & \frac{1}{2}\times(4x-y-2z) \\ 0 & 0 & 1 & \frac{1}{2}(-x+z) \end{pmatrix}$$

... in this next case RowReduce assumes that x+y-z is invertible.

$$\text{RowReduce}\left[\begin{pmatrix} 1 & 1 & 3 & x \\ 2 & 0 & 2 & y \\ 3 & 1 & 5 & z \end{pmatrix}\right] \text{ // MatrixForm}$$

$$\begin{pmatrix} 1 & 0 & 1 & 0 \\ 0 & 1 & 2 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

To correct this, SymbolicRR gives the type of answer one would arrive at when doing row reduction by hand.

$$\text{SymbolicRR}\left[\begin{pmatrix} 1 & 1 & 3 & x \\ 2 & 0 & 2 & y \\ 3 & 1 & 5 & z \end{pmatrix}, \{x, y, z\}\right] \text{ // MatrixForm}$$

$$\begin{pmatrix} 1 & 0 & 1 & \frac{y}{2} \\ 0 & 1 & 2 & x-\frac{y}{2} \\ 0 & 0 & 0 & -4(x+y-z) \end{pmatrix}$$

Note the importance of including the correct variables.

```
SymbolicRR[{{1, 1, 3, x}, {2, 0, 2, y}, {3, 1, 5, z}}, {a, b, c}] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 1 & 0 \\ 0 & 1 & 2 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

The variables are optional, so SymbolicRR can be used in place of RowReduce.

```
SymbolicRR[{{3, 5, 1}, {-2, 3, 8}, {3, 1, 9}}] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

SymbolicRR will attempt to make simplifications such as sin^2(x) + cos^2(x) = 1 and continue with row reduction.

```
SymbolicRR[{{1, 1, 2, Sin[x]^2}, {1, 1, 2, -Cos[x]^2 }, {7, 8, 9, z}}, {x, y, z}] //
 MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 7 & 0 \\ 0 & 1 & -5 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

This process of simplifying and continuing with row reduction will only occur 10 times, though that can be easily changed. This is to prevent *Mathematica* from entering an infinite loop when an incorrect matrix is inserted.

```
a = {{3, 4}, {5, 6}};
SymbolicRR[a, {x, y, z}] // MatrixForm
SymbolicRR[A, {x, y, z}] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

```
RowReduce[RowReduce[RowReduce[RowReduce[
    RowReduce[RowReduce[RowReduce[RowReduce[RowReduce[RowReduce[A, ZeroTest →
            (Or @@ Table[! FreeQ[#1, symb], {symb, {x, y, z}}] || PossibleZeroQ[#1] &)],
          ZeroTest → (Or @@ Table[! FreeQ[#1, symb], {symb, {x, y, z}}] ||
              PossibleZeroQ[#1] &)], ZeroTest →
          (Or @@ Table[! FreeQ[#1, symb], {symb, {x, y, z}}] || PossibleZeroQ[#1] &)],
        ZeroTest → (Or @@ Table[! FreeQ[#1, symb], {symb, {x, y, z}}] ||
            PossibleZeroQ[#1] &)], ZeroTest →
        (Or @@ Table[! FreeQ[#1, symb], {symb, {x, y, z}}] || PossibleZeroQ[#1] &)],
      ZeroTest → (Or @@ Table[! FreeQ[#1, symb], {symb, {x, y, z}}] ||
          PossibleZeroQ[#1] &)], ZeroTest →
      (Or @@ Table[! FreeQ[#1, symb], {symb, {x, y, z}}] || PossibleZeroQ[#1] &)],
    ZeroTest → (Or @@ Table[! FreeQ[#1, symb], {symb, {x, y, z}}] ||
        PossibleZeroQ[#1] &)], ZeroTest →
    (Or @@ Table[! FreeQ[#1, symb], {symb, {x, y, z}}] || PossibleZeroQ[#1] &)],
  ZeroTest → (Or @@ Table[! FreeQ[#1, symb], {symb, {x, y, z}}] ||
      PossibleZeroQ[#1] &)]
```

Documentation can be viewed using...

```
? SymbolicRR
```

SymbolicRR[*matrix*, {*var*$_1$, *var*$_2$, ...}] row reduces a matrix without assuming expressions involving variables are non–zero.